

# Software LINGO

Carlos Castro

UTFSM

September 2017

# Un problema de transporte

Origen	Producción
A	5.000
B	6.000
C	2.500

Destino	Demanda
I	6.000
II	4.000
III	2.000
IV	1.500

Costos de transporte unitarios:

		Destino			
		I	II	III	IV
Origen	A	3	2	7	6
	B	7	5	2	3
	C	2	5	4	5

# Modelo extendido

$$\text{MIN} = 3X_{11} + 2X_{12} + 7X_{13} + 6X_{14} + 7X_{21} + 5X_{22} + 2X_{23} + 3X_{24} \\ + 2X_{31} + 5X_{32} + 4X_{33} + 5X_{34};$$

$$X_{11} + X_{12} + X_{13} + X_{14} \leq 5000;$$

$$X_{21} + X_{22} + X_{23} + X_{24} \leq 6000;$$

$$X_{31} + X_{32} + X_{33} + X_{34} \leq 2500;$$

$$X_{11} + X_{21} + X_{31} \geq 6000;$$

$$X_{12} + X_{22} + X_{32} \geq 4000;$$

$$X_{13} + X_{23} + X_{33} \geq 2000;$$

$$X_{14} + X_{24} + X_{34} \geq 1500;$$

- Cada línea debe ser terminada por un punto y coma ;
- $<$  es equivalente a  $\leq$  y  $>$  es equivalente a  $\geq$
- Expresiones pueden ocupar varias líneas

```
MIN = 3X11 + 2X12 + 7X13 + 6X14 + 7X21 + 5X22  
      + 2X23 + 3X24 + 2X31 + 5X32 + 4X33 + 5X34;
```

- No se hace diferencia entre mayúsculas y minúsculas
- Identificadores: letra mayúscula (A-Z) o minúscula (a-z), seguida de letras, dígitos (0-9) o *underscore* (`_`)
- Largo máximo de identificadores: 32 caracteres

# El problema de transporte: Modelo general

- Variables de decisión:
- $x_{ij}$ : cantidad de unidades transportadas desde el nodo  $i$  al nodo  $j$ ;  $\forall i = 1, \dots, m, \forall j = 1, \dots, n$

- Función objetivo:

$$\text{Min } z = \sum_{i=1}^m \sum_{j=1}^n c_{ij} \times x_{ij}$$

- Restricciones:

- Capacidad de producción:

$$\sum_{j=1}^n x_{ij} \leq s_i; \quad \forall i = 1, \dots, m$$

- Satisfacción de demandas:

$$\sum_{i=1}^m x_{ij} \geq d_j; \quad \forall j = 1, \dots, n$$

- Valores posibles para las variables:

$$x_{ij} \geq 0; \quad \forall i = 1, \dots, m; j = 1, \dots, n$$

- Identificación objetos relacionados entre sí
- En un problema de transporte: orígenes, destinos y rutas
- Restricciones tipo se aplican a cada elemento dentro de un conjunto

# Sección de conjuntos: Sintaxis

- Inicio: palabra reservada `SETS` seguida de dos puntos (`:`) en una línea
- Fin: palabra reservada `ENDSETS` en una línea
- Entre estos dos delimitadores, se especifica cada uno de los conjuntos identificados en el modelo

# Sección de conjuntos: Ejemplo

SETS:

```
ORIGEN / O1 O2 O3 / : OFERTA;  
DESTINO / D1 D2 D3 D4 / : DEMANDA;  
ruta (ORIGEN, DESTINO) : COSTO, X;  
ENDSETS
```

- ORIGEN: O1, O2 y O3, cada uno teniendo asociado un atributo OFERTA
- DESTINO: D1, D2, D3 y D4, cada uno teniendo asociado un atributo DEMANDA
- RUTA: 12 rutas en la red de transporte, cada uno teniendo asociado atributos COSTO y X. Este conjunto se deriva de los conjuntos ORIGEN y DESTINO
- LINGO genera cada par ordenado (ORIGEN, DESTINO)
- 12 elementos miembros del conjunto RUTA: primer elemento O1D1, segundo O1D2 y así sucesivamente hasta el duodécimo elemento O3D4



Notación matemática	Sintaxis LINGO
$Max$	MAX =
$Min$	MIN =
$x_{ij}$	X(I,J)
$c_{ij}$	C(I,J)
$\times$	*
$\sum$	@SUM( : )
$\forall$	@FOR( : )

- Función objetivo

$$\text{Min } z = \sum_{i=1}^m \sum_{j=1}^n c_{ij} \times x_{ij}$$

es expresada en LINGO por:

```
MIN = @SUM( RUTA(I,J) : COSTO(I,J) * X(I,J) );
```

- Restricciones
  - Capacidad de producción

$$\sum_{j=1}^n x_{ij} \leq s_i; \quad \forall i = 1, \dots, m$$

es expresada en LINGO por:

```
@FOR (ORIGEN (I) : @SUM ( DESTINO (J) : X (I, J) ) <= OFERTA (I) );
```

- Satisfacción de demandas

$$\sum_{i=1}^m x_{ij} \geq d_j; \quad \forall j = 1, \dots, n$$

es expresada en LINGO por:

```
@FOR (DESTINO (J) : @SUM ( ORIGEN (I) : X (I, J) ) >= DEMANDA (J) );
```

- Inicio: palabra reservada `DATA` seguida de dos puntos (:) en una línea
- Fin: la palabra reservada `ENDDATA` en una línea
- Entre estos dos delimitadores, se especifica cada uno de los datos identificados en el modelo

# Sección de datos: ejemplo

DATA:

OFERTA = 5000 6000 2500;

DEMANDA = 6000 4000 2000 1500;

COSTO = 3 2 7 6

7 5 2 3

2 5 4 5;

ENDDATA

- Atributos OFERTA del conjunto ORIGEN: 5000, 6000 y 2500, respectivamente.
- Atributos DEMANDA del conjunto DESTINO: 6000, 4000, 2000 y 1500, respectivamente.
- Atributos COSTO del conjunto de dos dimensiones RUTA: 3, 2, 7, 6, 7, 5, 2, 3, 2, 5, 4 y 5, respectivamente
- LINGO incrementa primero el índice más externo: primer atributo  $COSTO(O1,D1)$  es 3, segundo  $COSTO(O1,D2)$  es 2 y así sucesivamente hasta  $COSTO(O3,D4)$  es 5

# Modelo completo

SETS:

ORIGEN / O1 O2 O3 / : OFERTA;

DESTINO / D1 D2 D3 D4 / : DEMANDA;

ruta (ORIGEN, DESTINO): COSTO, X;

ENDSETS

MIN = @SUM( ruta(I,J): COSTO(I,J) \* X(I,J));

@FOR( ORIGEN(I):

@SUM( DESTINO(J): X(I,J)) <= OFERTA(I));

@FOR( DESTINO(J):

@SUM( ORIGEN(I): X(I,J)) >= DEMANDA(J));

DATA:

OFERTA = 5000 6000 2500;

DEMANDA = 6000 4000 2000 1500;

COSTO = 3 2 7 6

7 5 2 3

2 5 4 5;

ENDDATA

END

# Modelo en extenso

```
MIN 3 X(O1,D1) + 2 X(O1,D2) + 7 X(O1,D3) + 6 X(O1,D4)
+ 7 X(O2,D1) + 5 X(O2,D2) + 2 X(O2,D3) + 3 X(O2,D4) +
2 X(O3,D1) + 5 X(O3,D2) + 4 X(O3,D3) + 5 X(O3,D4)
SUBJECT TO
2] X(O1,D1) + X(O1,D2) + X(O1,D3) + X(O1,D4) <= 5000
3] X(O2,D1) + X(O2,D2) + X(O2,D3) + X(O2,D4) <= 6000
4] X(O3,D1) + X(O3,D2) + X(O3,D3) + X(O3,D4) <= 2500
5] X(O1,D1) + X(O2,D1) + X(O3,D1) >= 6000
6] X(O1,D2) + X(O2,D2) + X(O3,D2) >= 4000
7] X(O1,D3) + X(O2,D3) + X(O3,D3) >= 2000
8] X(O1,D4) + X(O2,D4) + X(O3,D4) >= 1500
END
```

# Solución

Global optimal solution found at step: 7

Objective value: 39500.00

Variable Value Reduced Cost

OFERTA( O1) 5000.000 0.0000000

OFERTA( O2) 6000.000 0.0000000

OFERTA( O3) 2500.000 0.0000000

DEMANDA( D1) 6000.000 0.0000000

DEMANDA( D2) 4000.000 0.0000000

DEMANDA( D3) 2000.000 0.0000000

DEMANDA( D4) 1500.000 0.0000000

COSTO( O1, D1) 3.000000 0.0000000

COSTO( O1, D2) 2.000000 0.0000000

COSTO( O1, D3) 7.000000 0.0000000

COSTO( O1, D4) 6.000000 0.0000000

COSTO( O2, D1) 7.000000 0.0000000

COSTO( O2, D2) 5.000000 0.0000000

COSTO( O2, D3) 2.000000 0.0000000

COSTO( O2, D4) 3.000000 0.0000000

COSTO( O3, D1) 2.000000 0.0000000

COSTO( O3, D2) 5.000000 0.0000000

COSTO( O3, D3) 4.000000 0.0000000

COSTO( O3, D4) 5.000000 0.0000000



# Solución

```
X( O1, D1) 3500.000 0.0000000
X( O1, D2) 1500.000 0.0000000
X( O1, D3) 0.0000000 8.0000000
X( O1, D4) 0.0000000 6.0000000
X( O2, D1) 0.0000000 1.0000000
X( O2, D2) 2500.000 0.0000000
X( O2, D3) 2000.000 0.0000000
X( O2, D4) 1500.000 0.0000000
X( O3, D1) 2500.000 0.0000000
X( O3, D2) 0.0000000 4.0000000
X( O3, D3) 0.0000000 6.0000000
X( O3, D4) 0.0000000 6.0000000
```

```
Row Slack or Surplus Dual Price
1 39500.00 1.0000000
2 0.0000000 3.0000000
3 0.0000000 0.0000000
4 0.0000000 4.0000000
5 0.0000000 -6.0000000
6 0.0000000 -5.0000000
7 0.0000000 -2.0000000
8 0.0000000 -3.0000000
```

- Inicio: signo de exclamación !
- Fin: punto y coma ;
- Todo el texto entre estos dos delimitadores es considerado un comentario y, por lo tanto, ignorado por LINGO

```
!Capacidad origen uno;  
X11 + X12 + X13 + X14 <= 5000;
```

Un comentario puede ocupar más de una línea:

```
!Capacidad  
origen uno;  
X11 + X12 + X13 + X14 <= 5000;
```

Un comentario también puede compartir líneas con otras expresiones LINGO:

```
X11 + X12 + X13 + X14 <= 5000; !Capacidad origen uno;  
!Capacidad origen uno; X11 + X12 + X13 + X14 <= 5000;
```

Finalmente, un comentario puede estar incluso al interior de una expresión LINGO:

```
MIN = !Costo origen uno; 3X11 + 2X12 + 7X13 + 6X14 +  
!Costo origen dos; 7X21 + 5X22 + 2X23 + 3X24 +  
!Costo origen tres; 2X31 + 5X32 + 4X33 + 5X34;  
  
X = 1.5 * !Comentario al interior de restriccion; Y +  
Z / 2 * Y;
```

- Etiquetas para función objetivo y restricciones
- No son obligatorios
- Utilizados por LINGO en las soluciones generadas haciendo más fácil su interpretación
- Muchos mensajes de error de LINGO hacen referencia a las expresiones por su identificación

```
[OBJETIVO] MIN = @SUM ( RUTA(I,J) : COSTO(I,J) * X(I,J) );
```

asigna el identificador `OBJETIVO` a la fila correspondiente a la función objetivo

```
@FOR ( ORIGEN(I) : [FILA_DE_CAPACIDAD]  
@SUM ( DESTINO(J) : X(I,J) ) <= OFERTA(I) );
```

asigna el identificador `FILA_DE_CAPACIDAD` a las restricciones de capacidad de producción

- Inicio: palabra reservada `TITLE`
- Fin: punto y coma ;
- Todo el texto entre estos dos delimitadores es considerado como el título del modelo

```
TITLE Modelo de Transporte;
```



# Modelo completo documentado

MODEL:

TITLE Modelo de Transporte;

! Problema de transporte considerando 3 origenes  
y 4 destinos;

SETS:

ORIGEN / O1 O2 O3 /: OFERTA;

DESTINO / D1 D2 D3 D4 /: DEMANDA;

ruta (ORIGEN, DESTINO): COSTO, X;

ENDSETS

# Modelo completo documentado

```
! Funcion objetivo;
```

```
[OBJETIVO] MIN = @SUM ( RUTA(I,J): COSTO(I,J) * X(I,J) );
```

```
! Restricciones de capacidad;
```

```
@FOR ( ORIGEN(I): [FILA_DE_CAPACIDAD]
```

```
@SUM ( DESTINO(J): X(I,J) ) <= OFERTA(I) );
```

```
! Restricciones de demanda;
```

```
@FOR ( DESTINO(J): [FILA_DE_DEMANDA]
```

```
@SUM ( ORIGEN(I): X(I,J) ) >= DEMANDA(J) );
```

# Modelo completo documentado

```
! Datos;  
DATA:  
OFERTA = 5000 6000 2500;  
DEMANDA = 6000 4000 2000 1500;  
COSTO = 3 2 7 6  
7 5 2 3  
2 5 4 5;  
ENDDATA  
  
END
```

# Modelo en extenso documentado

```
MIN 3 X(O1,D1) + 2 X(O1,D2) + 7 X(O1,D3) + 6 X(O1,D4) + 7 X(O2,D1) + 5 X(O2,D2) + 2 X(O2,D3)
+ 3 X(O2,D4) + 2 X(O3,D1) + 5 X(O3,D2) + 4 X(O3,D3) + 5 X(O3,D4)
SUBJECT TO
FILA_DE_CAPACIDAD(O1) ] X(O1,D1)+ X(O1,D2) + X(O1,D3) + X(O1,D4) <= 5000
FILA_DE_CAPACIDAD(O2) ] X(O2,D1)+ X(O2,D2) + X(O2,D3) + X(O2,D4) <= 6000
FILA_DE_CAPACIDAD(O3) ] X(O3,D1)+ X(O3,D2) + X(O3,D3) + X(O3,D4) <= 2500
FILA_DE_DEMANDA(D1) ] X(O1,D1)+ X(O2,D1) + X(O3,D1) >= 6000
FILA_DE_DEMANDA(D2) ] X(O1,D2)+ X(O2,D2) + X(O3,D2) >= 4000
FILA_DE_DEMANDA(D3) ] X(O1,D3)+ X(O2,D3) + X(O3,D3) >= 2000
FILA_DE_DEMANDA(D4) ] X(O1,D4)+ X(O2,D4) + X(O3,D4) >= 1500
END
```

# Solución

Global optimal solution found at step: 7

Objective value: 39500.00

Model Title: Modelo de Transporte

Variable Value Reduced Cost

OFERTA( O1) 5000.000 0.0000000

OFERTA( O2) 6000.000 0.0000000

OFERTA( O3) 2500.000 0.0000000

DEMANDA( D1) 6000.000 0.0000000

DEMANDA( D2) 4000.000 0.0000000

DEMANDA( D3) 2000.000 0.0000000

DEMANDA( D4) 1500.000 0.0000000

COSTO( O1, D1) 3.000000 0.0000000

COSTO( O1, D2) 2.000000 0.0000000

COSTO( O1, D3) 7.000000 0.0000000

COSTO( O1, D4) 6.000000 0.0000000

COSTO( O2, D1) 7.000000 0.0000000

COSTO( O2, D2) 5.000000 0.0000000

COSTO( O2, D3) 2.000000 0.0000000

COSTO( O2, D4) 3.000000 0.0000000

COSTO( O3, D1) 2.000000 0.0000000

COSTO( O3, D2) 5.000000 0.0000000

COSTO( O3, D3) 4.000000 0.0000000

COSTO( O3, D4) 5.000000 0.0000000

# Solución

```
X( O1, D1) 3500.000 0.0000000
X( O1, D2) 1500.000 0.0000000
X( O1, D3) 0.0000000 8.0000000
X( O1, D4) 0.0000000 6.0000000
X( O2, D1) 0.0000000 1.0000000
X( O2, D2) 2500.000 0.0000000
X( O2, D3) 2000.000 0.0000000
X( O2, D4) 1500.000 0.0000000
X( O3, D1) 2500.000 0.0000000
X( O3, D2) 0.0000000 4.0000000
X( O3, D3) 0.0000000 6.0000000
X( O3, D4) 0.0000000 6.0000000
```

Row Slack or Surplus Dual Price

```
OBJETIVO 39500.00 1.0000000
FILA DE CAPACIDAD(O1) 0.0000000 3.0000000
FILA DE CAPACIDAD(O2) 0.0000000 0.0000000
FILA DE CAPACIDAD(O3) 0.0000000 4.0000000
FILA DE DEMANDA(D1) 0.0000000 -6.0000000
FILA DE DEMANDA(D2) 0.0000000 -5.0000000
FILA DE DEMANDA(D3) 0.0000000 -2.0000000
FILA DE DEMANDA(D4) 0.0000000 -3.0000000
```

# Tamaño de los modelos

Versión	Total de variables	Variables enteras	Variables no lineales	Restricciones
Demo/Web	300	30	30	150
Solver Suite	500	50	50	250
Super	2.000	200	200	1.000
Hyper	8.000	800	800	4.000
Industrial	32.000	3.200	3.200	16.000
Extended	Ilimitada	Ilimitada	Ilimitada	Ilimitada

# Software LINGO

Carlos Castro

UTFSM

September 2017